Pensieve header: The naive Kh Program - a variant of the program I first wrote in Kyoto in September 2001.

tex

```
\documentclass[11pt,notitlepage]{article}
\usepackage{dbnsymb, amsmath, amssymb, multicol, stmaryrd, pifont,
 amscd, colortbl, mathtools, wasysym, needspace, import, longtable,
 overpic, bbm, pdfpages, array, blkarray, setspace, graphicx, tensor}
\usepackage{txfonts}      % for the likes of \coloneqq.
\usepackage[usenames,dvipsnames]{xcolor}
\usepackage[textwidth=8in,textheight=10.5in,centering]{geometry}
\usepackage[setpagesize=false]{hyperref} % Following http://tex.stackexchange.com/a/847/22475
\hypersetup{colorlinks, linkcolor={blue!50!black}, citecolor={blue!50!black}, urlcolor={blue!50!black}}
\usepackage[export]{adjustbox}

\def\nbpdfInput#1{\vskip 1mm\par\noindent\includegraphics[max width=\linewidth]{#1}}
\def\nbpdfEcho#1{\vskip 1mm\par\noindent\includegraphics{#1}}
\def\nbpdfPrint#1{\vskip 1mm\par\noindent\includegraphics{#1}}
\def\nbpdfText#1{\vskip 1mm\par\noindent\includegraphics{#1}}
\def\nbpdfMessage#1{}
\def\nbpdfOutput#1{\vskip 1mm\par\noindent\includegraphics[max width=\linewidth]{#1}}
\def\nbpdfSubsection#1{\vskip 1mm\par\noindent\includegraphics{#1}}
\def\nbpdfSubsubsection#1{\vskip 1mm\par\noindent\includegraphics{#1}}
\def\nbpdfgraphInput#1{\vskip 1mm\par\noindent\includegraphics{#1}}
\def\nbpdfgraphOutput#1{\vskip 1mm\par\noindent\includegraphics[width=1.5in]{#1}}

\pagestyle{empty} \parindent 0in
\begin{document}
\setlength{\abovedisplayskip}{0.5ex} \setlength{\belowdisplayskip}{0.5ex}
\setlength{\abovedisplayshortskip}{0ex} \setlength{\belowdisplayshortskip}{0ex}

\def\myurl{http://www.math.toronto.edu/~drorbn}
\href{\myurl}{Dror Bar-Natan}:
\href{\myurl/classes}{Classes}:
\href{\myurl/classes/\#2223}{2022-23}:
\href{\myurl/classes/23-FastComputations}{Fast Computations in Knot Theory}:
{\bf A Naive Khovanov Homology Program}

\begin{multicols*}{2}
```

pdf

We load a knot theory package only for pre-loaded PD data and for comparisons with known KH results:

pdf

```
Once[<< KnotTheory`]
```

pdf

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

pdf

```
pd = PD[Knot[4, 1]]
```

pdf

```
KnotTheory: Loading precomputed data in PD4Knots`.
```

pdf

```
PD[X[4, 2, 5, 1], X[8, 6, 1, 5], X[6, 3, 7, 4], X[2, 7, 3, 8]]
```

pdf

```
n₊[pd_PD] := Count[pd, X[i_, j_, k_, l_] /; j - l == 1 ∨ l - j > 1];
n₋[pd_PD] := Count[pd, X[i_, j_, k_, l_] /; l - j == 1 ∨ j - l > 1];
```

pdf

```
{n₊[pd], n₋[pd]}
```

pdf

```
{2, 2}
```

pdf

```
SetAttributes[p, Orderless]
```

pdf

```
S[pd_PD, s_String] := S[pd, Characters[s] /. {"0" → 0, "1" → 1, "*" → ⋆}]
```

pdf

```
S[pd, "01*0"]
```

pdf

```
S[PD[X[4, 2, 5, 1], X[8, 6, 1, 5], X[6, 3, 7, 4], X[2, 7, 3, 8]], {0, 1, ⋆, 0}]
```

pdf

```
m_ ⋄ n_ := Min[m, n];
S[pd_PD, a_List] := Times @@ ({List @@ pd, a}ᵀ /. {
         {X[i_, j_, k_, l_], 0} :> p[i, j]_(i⋄j) p[k, l]_(k⋄l),
         {X[i_, j_, k_, l_], 1} :> p[i, l]_(i⋄l) p[j, k]_(j⋄k),
         {x_X, ⋆} :> x}
     ) //. {
     p[i_, j_]_m_ p[j_, k_]_n_ :> p[i, k]_(m⋄n)
    } //. {
    X[i_, j_, k_, l_] p[i_, j_]_m_ p[k_, l_]_n_ :> (c_m c_n → c_(m⋄n)),
    X[i_, j_, k_, l_] p[i_, l_]_m_ p[j_, k_]_n_ :> (c_(m⋄n) → c_m c_n)
   } //. p[__]_m_ :> c_m
```

pdf

```
{S[pd, "0100"], S[pd, "0110"], S[pd, "01*0"]}
```

pdf

$$\{c_1\, c_2,\ c_1,\ c_1\, c_2 \to c_1\}$$

pdf

```
{S[pd, "0000"], S[pd, "1000"], S[pd, "*000"]}
```

pdf

$$\{c_1\, c_2\, c_3,\ c_1\, c_3,\ (c_1\, c_2 \to c_1)\ c_3\}$$

pdf

```
{S[pd, "1100"], S[pd, "1110"], S[pd, "11*0"]}
```

pdf

$$\{c_1,\ c_1\, c_2,\ c_1 \to c_1\, c_2\}$$

pdf

```
V[pd_PD, a_] := List @@ Expand[S[pd, a] /. c_x_ :> (vp_x + vm_x)]
```

pdf

```
V[pd, "0100"]
```

pdf

$$\{vm_1\, vm_2,\ vm_2\, vp_1,\ vm_1\, vp_2,\ vp_1\, vp_2\}$$

pdf

```
d[pd_PD, a_] := S[pd, a] /. {
    (c_x_ c_y_ → c_z_) * _. :> {vp_x vp_y → vp_z, vp_x vm_y → vm_z, vm_x vp_y → vm_z, vm_x vm_y → 0},
    (c_z_ → c_x_ c_y_) * _. :> {vp_z → vp_x vm_y + vm_x vp_y, vm_z → vm_x vm_y}
   }
```

pdf

```
{S[pd, "01*0"], d[pd, "01*0"]}
```

pdf

$$\{c_1\, c_2 \to c_1,\ \{vp_1\, vp_2 \to vp_1,\ vm_2\, vp_1 \to vm_1,\ vm_1\, vp_2 \to vm_1,\ vm_1\, vm_2 \to 0\}\}$$

```
{S[pd, "*000"], d[pd, "*000"]}
```

```
{S[pd, "11*0"], d[pd, "11*0"]}
```

```
{S[pd, "*000"] , V[pd, "0000"], d[pd, "*000"]}
```

$$\{(c_1\, c_2 \to c_1)\ c_3,\ \{vm_1\, vm_2\, vm_3,\ vm_2\, vm_3\, vp_1,\ vm_1\, vm_3\, vp_2,\ vm_3\, vp_1\, vp_2,\ vm_1\, vm_2\, vp_3,\ vm_2\, vp_1\, vp_3,\ vm_1\, vp_2\, vp_3,\ vp_1\, vp_2\, vp_3\},\ \{vp_1\, vp_2 \to vp_1,\ vm_2\, vp_1 \to vm_1,\ vm_1\, vp_2 \to vm_1,\ vm_1\, vm_2 \to 0\}\}$$

pdf

```
{V[pd, "0000"], V[pd, "0000"] /. d[pd, "*000"]}
```

pdf

$$\{\{vm_1\, vm_2\, vm_3,\ vm_2\, vm_3\, vp_1,\ vm_1\, vm_3\, vp_2,\ vm_3\, vp_1\, vp_2,\ vm_1\, vm_2\, vp_3,\ vm_2\, vp_1\, vp_3,\ vm_1\, vp_2\, vp_3,\ vp_1\, vp_2\, vp_3\},\ \{0,\ vm_1\, vm_3,\ vm_1\, vm_3,\ vm_3\, vp_1,\ 0,\ vm_1\, vp_3,\ vm_1\, vp_3,\ vp_1\, vp_3\}\}$$

pdf

```
udeg[P_] := Exponent[P /. {v_a_ :> q^Total[a], vp_ → q, vm_ → q^-1}, q]
```

pdf

```
udeg[v_{0,0,0,1,1} vm_1]
```

pdf

```
1
```

pdf

```
KC[pd_PD, r_] := If[r < -n_-[pd] || r > n_+[pd], {},
   Join @@
    (((v_#) V[pd, #]) & /@ Permutations[Table[0, n_+[pd] - r] ~Join~ Table[1, r + n_-[pd]]])
  ];
KC[pd_PD, r_, deg_] := Cases[KC[pd, r], u_ /; udeg[u] - 2 n_-[pd] + n_+[pd] == deg]
```

pdf

```
KC[pd, 1]
```

pdf

$$\{v_{\{0,1,1,1\}}\, vm_1\, vm_3,\ v_{\{0,1,1,1\}}\, vm_3\, vp_1,\ v_{\{0,1,1,1\}}\, vm_1\, vp_3,\ v_{\{0,1,1,1\}}\, vp_1\, vp_3,$$
$$v_{\{1,0,1,1\}}\, vm_1\, vm_3,\ v_{\{1,0,1,1\}}\, vm_3\, vp_1,\ v_{\{1,0,1,1\}}\, vm_1\, vp_3,\ v_{\{1,0,1,1\}}\, vp_1\, vp_3,$$
$$v_{\{1,1,0,1\}}\, vm_1\, vm_2,\ v_{\{1,1,0,1\}}\, vm_2\, vp_1,\ v_{\{1,1,0,1\}}\, vm_1\, vp_2,\ v_{\{1,1,0,1\}}\, vp_1\, vp_2,$$
$$v_{\{1,1,1,0\}}\, vm_1\, vm_2,\ v_{\{1,1,1,0\}}\, vm_2\, vp_1,\ v_{\{1,1,1,0\}}\, vm_1\, vp_2,\ v_{\{1,1,1,0\}}\, vp_1\, vp_2\}$$

pdf

```
KC[pd, 1, 1]
```

pdf

$$\{v_{\{0,1,1,1\}}\, vm_3\, vp_1,\ v_{\{0,1,1,1\}}\, vm_1\, vp_3,\ v_{\{1,0,1,1\}}\, vm_3\, vp_1,\ v_{\{1,0,1,1\}}\, vm_1\, vp_3,$$
$$v_{\{1,1,0,1\}}\, vm_2\, vp_1,\ v_{\{1,1,0,1\}}\, vm_1\, vp_2,\ v_{\{1,1,1,0\}}\, vm_2\, vp_1,\ v_{\{1,1,1,0\}}\, vm_1\, vp_2\}$$

pdf

```
dd[pd_PD][expr_] := Expand[expr] /. s_ * v_{a_} :> Expand[σ = 1;
    Sum[
     If[a[[i]] == 0, σ * v_{ReplacePart[a,1,i]} * s /. d[pd, List @@ ReplacePart[a, ⋆, i]], σ *= -1;
      0], {i, Length[a]}
     ]
    ]
```

pdf

```
KC[pd, 0, 1] // dd[pd]
```

pdf

$$\{v_{\{0,1,1,1\}}\, vm_3\, vp_1 + v_{\{1,0,1,1\}}\, vm_3\, vp_1,$$
$$v_{\{0,1,1,1\}}\, vm_1\, vp_3 + v_{\{1,0,1,1\}}\, vm_1\, vp_3,\ v_{\{0,1,1,1\}}\, vm_1\, vp_3 + v_{\{1,0,1,1\}}\, vm_1\, vp_3,$$
$$v_{\{1,1,0,1\}}\, vm_2\, vp_1 - v_{\{0,1,1,1\}}\, vm_3\, vp_1 + v_{\{1,1,0,1\}}\, vm_1\, vp_2 - v_{\{0,1,1,1\}}\, vm_1\, vp_3,$$
$$v_{\{1,1,1,0\}}\, vm_2\, vp_1 + v_{\{0,1,1,1\}}\, vm_3\, vp_1 + v_{\{1,1,1,0\}}\, vm_1\, vp_2 + v_{\{0,1,1,1\}}\, vm_1\, vp_3,$$
$$-v_{\{1,1,0,1\}}\, vm_2\, vp_1 - v_{\{1,0,1,1\}}\, vm_3\, vp_1 - v_{\{1,1,0,1\}}\, vm_1\, vp_2 - v_{\{1,0,1,1\}}\, vm_1\, vp_3,$$
$$-v_{\{1,1,1,0\}}\, vm_2\, vp_1 + v_{\{1,0,1,1\}}\, vm_3\, vp_1 - v_{\{1,1,1,0\}}\, vm_1\, vp_2 + v_{\{1,0,1,1\}}\, vm_1\, vp_3,$$
$$v_{\{1,1,0,1\}}\, vm_2\, vp_1 + v_{\{1,1,1,0\}}\, vm_2\, vp_1 + v_{\{1,1,0,1\}}\, vm_1\, vp_2 + v_{\{1,1,1,0\}}\, vm_1\, vp_2\}$$

```
KC[pd, 0, 1] // dd[pd] // dd[pd]
```

$$\{0, 0, 0, 0, 0, 0, 0, 0\}$$

pdf

```
Rank[pd_PD, r_, deg_] := (
    B0 = KC[pd, r, deg];
    B1 = KC[pd, r + 1, deg];
    If[B0 == {} ∨ B1 == {}, 0,
     dB0 = dd[pd][B0];
     MatrixRank[Table[Coefficient[db0, b1], {db0, dB0}, {b1, B1}]]
    ]
   );
```

pdf

```
Rank[pd, 1, 1]
```

pdf

```
3
```

pdf

```
Betti[pd_PD, r_, deg_] := Length[KC[pd, r, deg]] - Rank[pd, r, deg] - Rank[pd, r - 1, deg]
```

pdf

```
Betti[pd, 1, 1]
```

pdf

```
1
```

pdf

```
Kh1[pd_PD] := Sum[
  t^r q^deg Betti[pd, r, deg],
  {r, -n_-[pd], n_+[pd]},
  {deg, Union[udeg /@ KC[pd, r]] - 2 n_-[pd] + n_+[pd]}
 ]
```

pdf

```
Kh1[pd]
```

pdf

$$\frac{1}{q}+q+\frac{1}{q^5\,t^2}+\frac{1}{q\,t}+q\,t+q^5\,t^2$$

```
Timing@Table[
  K → Kh[K][q, t] == Kh1[PD@K],
  {K, AllKnots[{3, 6}]}
 ]
```

KnotTheory: Loading precomputed data in Kh4Knots`.

```
{7.23438, {Knot[3, 1] → True, Knot[4, 1] → True, Knot[5, 1] → True,
  Knot[5, 2] → True, Knot[6, 1] → True, Knot[6, 2] → True, Knot[6, 3] → True}}
```

tex

```
\end{multicols*}
```

exec

```
nb2tex$PDFWidth = 8.5;
```

pdf

```
m_ ⋄ n_ := Min[m, n];
```

```
Kh2[K_] := Module[{pd, np, nm, p, S, a, ⋆, c, V,
   vp, vm, d, udeg, KC, v, dd, σ, Rank, B0, B1, dB0, db0, b1, Betti},
  pd = PD[K];
  np = Count[pd, X[i_, j_, k_, l_] /; j - l == 1 ∨ l - j > 1];
  nm = Count[pd, X[i_, j_, k_, l_] /; l - j == 1 ∨ j - l > 1];
  SetAttributes[p, Orderless];
  S[a_List] := S[a] = Times @@ ({List @@ pd, a}ᵀ /. {
          {X[i_, j_, k_, l_], 0} :> p[i, j]ᵢ⋄ⱼ p[k, l]ₖ⋄ₗ,
          {X[i_, j_, k_, l_], 1} :> p[i, l]ᵢ⋄ₗ p[j, k]ⱼ⋄ₖ,
          {x_X, ⋆} :> x}
       ) //. {
       p[i_, j_]ₘ_ p[j_, k_]ₙ_ :> p[i, k]ₘ⋄ₙ
      } //. {
      X[i_, j_, k_, l_] p[i_, j_]ₘ_ p[k_, l_]ₙ_ :> (cₘ cₙ → c_{m⋄n}),
      X[i_, j_, k_, l_] p[i_, l_]ₘ_ p[j_, k_]ₙ_ :> (c_{m⋄n} → cₘ cₙ)
     } //. p[__]_{m}^· :> cₘ;
  V[a_] := V[a] = List @@ Expand[S[a] /. c_x_ :> (vp_x + vm_x)];
  d[a_] := d[a] = S[a] /. {
     (c_x_ c_y_ → c_z_) * _. :> {vp_x vp_y → vp_z, vp_x vm_y → vm_z, vm_x vp_y → vm_z, vm_x vm_y → 0},
     (c_z_ → c_x_ c_y_) * _. :> {vp_z → vp_x vm_y + vm_x vp_y, vm_z → vm_x vm_y}
    };
  udeg[P_] := Exponent[P /. {v_a_ :> q^Total[a], vp_ → q, vm_ → q^-1}, q];
  KC[r_] := KC[r] = If[r < -nm || r > np, {},
    Join @@ (((v_#) V[#]) & /@ Permutations[Table[0, np - r] ~Join~ Table[1, r + nm]])
   ];
  KC[r_, deg_] := KC[r, deg] = Cases[KC[r], u_ /; udeg[u] - 2 nm + np == deg];
  dd[expr_] := Expand[expr] /. s_ * v_a_ :> Expand[σ = 1;
    Sum[
     If[a⟦i⟧ == 0, σ * v_{ReplacePart[a,1,i]} * s /. d[List @@ ReplacePart[a, ⋆, i]], σ *= -1;
      0], {i, Length[a]}
     ]
    ];
  Rank[r_, deg_] := Rank[r, deg] = (
    B0 = KC[r, deg];
    B1 = KC[r + 1, deg];
    If[B0 == {} ∨ B1 == {}, 0,
     dB0 = dd[B0];
     MatrixRank[Table[Coefficient[db0, b1], {db0, dB0}, {b1, B1}]]
    ]
   );
  Betti[r_, deg_] := Length[KC[r, deg]] - Rank[r, deg] - Rank[r - 1, deg];
  Sum[
   t^r q^deg Betti[r, deg],
   {r, -nm, np},
```

```
    {deg, Union[udeg /@ KC[r]] - 2 nm + np}
   ]
  ]
```

exec

```
nb2tex$PDFWidth = 4.5;
```

tex

\begin{multicols*}{2}

pdf

```
Kh2[Knot[4, 1]]
```

pdf

$$\frac{1}{q}+q+\frac{1}{q^5\,t^2}+\frac{1}{q\,t}+q\,t+q^5\,t^2$$

pdf

```
Timing@Table[
  K → Kh[K][q, t] == Kh2[K],
  {K, AllKnots[{3, 6}]}
 ]
```

pdf

```
{0.90625, {Knot[3, 1] → True, Knot[4, 1] → True, Knot[5, 1] → True,
  Knot[5, 2] → True, Knot[6, 1] → True, Knot[6, 2] → True, Knot[6, 3] → True}}
```

pdf

```
Timing[kh12 = {Kh2[Knot[5, 1]], Kh2[Knot[10, 132]]}]
```

pdf

$$\left\{115.047,\ \left\{\frac{1}{q^5}+\frac{1}{q^3}+\frac{1}{q^{15}\,t^5}+\frac{1}{q^{11}\,t^4}+\frac{1}{q^{11}\,t^3}+\frac{1}{q^7\,t^2},\right.\right.$$
$$\left.\left.\frac{1}{q^3}+\frac{1}{q}+\frac{1}{q^{15}\,t^7}+\frac{1}{q^{11}\,t^6}+\frac{1}{q^{11}\,t^5}+\frac{1}{q^9\,t^4}+\frac{1}{q^7\,t^4}+\frac{1}{q^9\,t^3}+\frac{1}{q^5\,t^3}+\frac{2}{q^5\,t^2}+\frac{1}{q\,t}\right\}\right\}$$

pdf

```
Expand@Cancel[kh12 /. t → -1 / (q + q^-1)]
```

pdf

$$\left\{-\frac{1}{q^{14}}+\frac{1}{q^{12}}-\frac{1}{q^{10}}+\frac{1}{q^8}+\frac{1}{q^4},\ -\frac{1}{q^{14}}+\frac{1}{q^{12}}-\frac{1}{q^{10}}+\frac{1}{q^8}+\frac{1}{q^4}\right\}$$

pdf

```
{Jones[Knot[5, 1]][q^2], Jones[Knot[10, 132]][q^2]}
```

pdf

$$\left\{-\frac{1}{q^{14}}+\frac{1}{q^{12}}-\frac{1}{q^{10}}+\frac{1}{q^8}+\frac{1}{q^4},\ -\frac{1}{q^{14}}+\frac{1}{q^{12}}-\frac{1}{q^{10}}+\frac{1}{q^8}+\frac{1}{q^4}\right\}$$

tex

\end{multicols*}
\end{document} \endinput